# Finding the area enclosed by an implicit function [closed]

How can I solve area of following plot by using double integral?

• Next time: Want an area?, show a region, this is a curve. Want an area using a specific method?, where are you stuck in implementation then? Write a formula at least. This is your third image-only question, they are not welcomed here, though answered sometimes, because people can't copy code/formulas from an image. – Kuba May 8 '17 at 8:28

You can convert to polar coordinates:

r[t_] := 2 Cos[t]^2
Integrate[r[t]^2/2, {t, 0, 2 Pi}]


yields $3\pi/2\approx 4.71239$

or you can use Green's Theorem (with $\vec{F}=\{-y/2,x/2\}$):=

Integrate[{-r[t] Sin[t], r[t] Cos[t]} .D[{r[t] Cos[t], r[t] Sin[t]}, t]],{t,0,2Pi}]/2


also yielding $3\pi/2$

or approximate using ImplicitRegion:

reg = ImplicitRegion[(x^2 + y^2)^3 <= 4 x^4, {{x, -2, 2}, {y, -2, 2}}]
RegionMeasure[DiscretizeRegion[reg, MaxCellMeasure -> {"Length" -> 0.01}]]


yields: 4.71238

See Kuba comment below for shorter ImplicitRegion solution:

Area@ImplicitRegion[(x^2. + y^2)^3. <= 4. x^4., {x, y}]


yields: 4.71239

• +1 Finite precision numbers will shorten ImplicitRegion solution: Area @ ImplicitRegion[(x^2. + y^2)^3. <= 4. x^4., {x, y}] – Kuba May 8 '17 at 8:14